

IDENTIFICATION

PRODUCT CODE: Digital-8-18-U-Sym
PRODUCT NAME: Alphanumeric Message Typeout
DATE CREATED: February 16, 1967
MAINTAINER: Software Service Group

Digital-8-18-U-Sym

1. ABSTRACT

A basic subroutine to type messages packed in computer words. Two 6-bit characters are packed internally in a single word. All ASR-33 codes from 301 to 337 and from 240 to 277 (excepting 243 and 245) can be typed. The typing of line-feed (code 212) and carriage-return (code 215) are made possible by arbitrarily assigning internal codes of 43 and 45, respectively, to represent these characters, thus preventing the output of ASCII codes 243 (#) and 245 (%).

2. REQUIREMENTS

Storage

This subroutine occupies 48 (decimal) storage locations in core plus autoindex register 10 (octal) on page 0.

Equipment

Basic PDP-8

3. USAGE

3.1 Loading

This subroutine may be placed in memory by the use of the Binary Loader. The library tape supplied is symbolic.

3.2 Calling Sequence

The calling sequence is designed so that the user may easily incorporate messages in his program. The following example illustrates a simple usage of this program.

```
/TEST 8-18-U
/START ADDRESS IS 400
/PRINTS AB HI E.R. DOW AND HALTS AT LOC. 407
*400          /ORIGIN AT 400
START,       JMS I ADDR/  JMS TO MESSAGE SUBR
              0102        /AB
              1011        /HI
              0556        /E.
              2256        /R.
              0417        /DO
              2700        /W+ END CHAR
END,         HLT          /END OF PROGRAM
ADDR,       MESSAGE     /ADDRESS OF MESSAGE TYPEOUT SUBROUTINE
```

4. RESTRICTIONS

The end-of-message code consists of 00_8 or (000000_2) , of course, only an end-of-message code may appear in the most significant six bits of location at the end of the message if the message consists of an even number of characters.

5. DESCRIPTION

The ASCII code breaks down into two main groups: first, the set of codes from 301 to 337 inclusive; and second, the set of codes from 240 to 277 inclusive. Combined, these two sets represent 63 characters. It is not necessary to store the most significant octal digit of the code for these characters, 2 or 3, internally since it may be computed from a knowledge of the least two significant digits, in other words from a "stripped" code.

Codes 3XX where digits XX are greater than 37 and codes 2XX where XX is less than 40 do not fit into this scheme and must be handled by special means. Only two such codes are necessary to accomplish the purposes of this subroutine. These are line-feed (code 212) and carriage-return (code 215) which are assigned "stripped" code representations of 43 and 45, respectively, making the actual codes 243 (#) and 245 (%) illegal for this subroutine; e.g., when an internal code of 43 is found, it is discarded and a 212 is sent to the ASR-33.

Section 7 contains a complete table of internal and external codes legal and illegal for this subroutine.

6. METHOD

Upon entry MESSAGE will hold the address of the first message word. One is subtracted from this and the result deposited in autoindex register 10. The main loop is then entered.

The message word is deposited in MSRGHT, then rotated six bits to the right. A jump to the minor subroutine TYPECH causes the character now contained in the six least significant bits of C(AC) to be typed. A second jump to TYPECH causes the character contained in the least significant six bits of MSRGHT to be typed.

If at any time TYPECH finds the least significant six bits of its current data word to be 0 (the end-of-message code), MESSAGE will return to the calling program.

After each two passes through TYPECH (if an end-of-message code is not encountered), the next message word is picked up by an indirect TAD instruction referencing location 0010 (octal) and the main loop repeats.

7. FORMAT

For this program external (ASCII) and internal core formats may best be illustrated by the following tables.

LEGAL CHARACTERS

| Internal (Stripped) | ASCII | Character | Internal (Stripped) | ASCII | Character |
|---------------------|-------|-----------|---------------------|-------|-----------|
| 01 | 301 | A | 12 | 312 | J |
| 02 | 302 | B | 13 | 313 | K |
| 03 | 303 | C | 14 | 314 | L |
| 04 | 304 | D | 15 | 315 | M |
| 05 | 305 | E | 16 | 316 | N |
| 06 | 306 | F | 17 | 317 | O |
| 07 | 307 | G | 20 | 320 | P |
| 10 | 310 | H | 21 | 321 | Q |
| 11 | 311 | I | 22 | 322 | R |

LEGAL CHARACTERS (continued)

| Internal (Stripped) | ASCII | Character | Internal (Stripped) | ASCII | Character |
|------------------------|-------|-----------------|------------------------|-------|-----------|
| 23 | 323 | S | 52 | 252 | * |
| 24 | 324 | T | 53 | 253 | + |
| 25 | 325 | U | 54 | 254 | , |
| 26 | 326 | V | 55 | 255 | - |
| 27 | 327 | W | 56 | 256 | . |
| 30 | 330 | X | 57 | 257 | \ |
| 31 | 331 | Y | 60 | 260 | 0 |
| 32 | 332 | Z | 61 | 261 | 1 |
| 33 | 333 | [| 62 | 262 | 2 |
| 34 | 334 | / | 63 | 263 | 3 |
| 35 | 335 |] | 64 | 264 | 4 |
| 36 | 336 | ↑ | 65 | 265 | 5 |
| 37 | 337 | → | 66 | 266 | 6 |
| 40 | 240 | space | 67 | 267 | 7 |
| 41 | 241 | : | 70 | 270 | 8 |
| 42 | 242 | " | 71 | 271 | 9 |
| 43 | 243 | line feed | 72 | 272 | : |
| 44 | 244 | \$ | 73 | 273 | ; |
| 45 | 245 | carriage return | 74 | 274 | < |
| 46 | 246 | & | 75 | 275 | = |
| 47 | 247 | ' | 76 | 276 | > |
| 50 | 250 | (| 77 | 277 | ? |
| 51 | 251 |) | | | |

Illegal codes, that is codes that will never be sent to the ASR-33 by this subroutine, are shown in the next table. The characters represented by these ASCII codes cannot be typed by this subroutine.

ILLEGAL CODES

| ASCII | Character | Reason for Illegality |
|-------|-----------|---|
| 300 | @ | Stripped code 00 needed for end-of-message code |
| 374 | ACK | Greater than 37 internally |
| 375 | ALT MODE | Greater than 37 internally |
| 377 | RUB OUT | Greater than 37 internally |
| 204 | EOT | Less than 40 internally |
| 205 | W RU | Less than 40 internally |
| 206 | RU | Less than 40 internally |
| 207 | BELL | Less than 40 internally |
| 243 | # | Arbitrarily used in stripped form for FORM FEED |
| 245 | % | Arbitrarily used in stripped form for CAR. RETURN |

Digital-8-18-U-Sym

Note that there are only ten illegal codes. The illegal codes are not frequently necessary in alphanumeric messages. Furthermore, in cases where they might be useful they can be represented by simple combinations of legal characters. For example @ may be represented by "AT" while # may commonly be represented by "NO."

8. EXECUTION TIME

This subroutine is output limited.

9. PROGRAM LISTING

```

/DIGITAL 8-18-U
/MESSAGE TYPE-OUT
/CALL WITH A JMS MESSAGE
/WITH DATA FOLLOWING
/RETURN FOLLOWING END OF MESSAGE
/COE(00)

```

```

0200 0000 MESSAGE, 0
0201 7240 CLA CMA /SET C(AC)=-1
0202 1200 TAD MESSAGE /ADD LOCATION
0203 3010 DCA 10 /AUTO-INDEX REGISTER
0204 1410 TAD I 10 /FETCH FIRST WORD
0205 3216 DCA MSRGT /SAVE IT
0206 1216 TAD MSRGT
0207 7012 RTR
0210 7012 RTR /ROTATE 6 BITS RIGHT
0211 7012 RTR
0212 4217 JMS TYPECH /TYPE IT
0213 1216 TAD MSRGT /GET DATA AGAIN
0214 4217 JMS TYPECH /TYPE RIGHT HALF
0215 5204 JMP MESSAGE+4 /CONTINUE
0216 0000 MSRGT, 0 /TEMPORARY STORAGE

0217 0000 TYPECH, 0 /TYPE CHARACTER IN C(AC)6-11
0220 0250 AND MASK77
0221 7450 SNA /IS IT END OF MESSAGE?
0222 5410 JMP I 10 /YES: EXIT
0223 1251 TAD M40 /SUBTRACT 40
0224 7500 SMA /<40?
0225 5230 JMP .+3 /NO
0226 1252 TAD C340 /YES: ADD 300
0227 5243 JMP MTP /TO CODES <40
0230 1253 TAD M3 /SUBTRACT 3
0231 7440 SZA /IS IT ZERO?
0232 5235 JMP .+3 /NO
0233 1254 TAD C212 /YES: CODE 43 IS
0234 5243 JMP MTP /LINE-FEED (212)
0235 1255 TAD M2 /SUBTRACT 2
0236 7440 SZA /IS IT ZERO?
0237 5242 JMP .+3 /NO
0240 1256 TAD C215 /YES: CODE 45 IS
0241 5243 JMP MTP /CARRIAGE-RETURN (215)

```

Digital-8-18-U-Sym

| | | | | |
|------|------|------|--------------|------------------------|
| 0242 | 1257 | | TAD C245 | /ADD 200 TO OTHERS >40 |
| 0243 | 6046 | MTP, | TL5 | /TRANSMIT CHARACTER |
| 0244 | 6041 | | TSF | /WAIT FOR FLAG |
| 0245 | 5244 | | JMP .-1 | /NOT SET YET |
| 0246 | 7200 | | CLA | /SET: CLEAR C(AC) |
| 0247 | 5617 | | JMP I TYPECH | /RETURN |

/CONSTANTS

| | | | |
|------|------|---------|-----|
| 0250 | 0077 | MASK77, | 77 |
| 0251 | 7740 | M40, | -40 |
| 0252 | 0340 | C340, | 340 |
| 0253 | 7775 | M3, | -3 |
| 0254 | 0212 | C212, | 212 |
| 0255 | 7776 | M2, | -2 |
| 0256 | 0215 | C215, | 215 |
| 0257 | 0245 | C245, | 245 |

| | |
|---------|------|
| C212 | 0254 |
| C215 | 0256 |
| C245 | 0257 |
| C340 | 0252 |
| MASK77 | 0250 |
| MESSAGE | 0200 |
| MSRGHT | 0216 |
| MTP | 0243 |
| M2 | 0255 |
| M3 | 0253 |
| M40 | 0251 |
| TYPECH | 0217 |

10. REFERENCES

Digital-8-19-U (Teletype Output Subroutines) and Digital-8-20-U (Character String Typeout).

